


Обработка исключений



Рассматриваемые вопросы

- Определение исключений PL/SQL
- Распознавание необработанных исключений
- Различные типы обработчиков исключений PL/SQL и их использование
- Перехват непредусмотренных исключений
- Эффект распространения исключения на вызывающую среду во вложенных блоках
- Настройка сообщений об исключениях PL/SQL

Пример

```
SET SERVEROUTPUT ON
DECLARE
  lname VARCHAR2(15);
BEGIN
  SELECT last_name INTO lname FROM employees WHERE
  first_name='John';
  DBMS_OUTPUT.PUT_LINE ('John''s last name is : '
  ||lname);
END;
/
```


Пример

```
SET SERVEROUTPUT ON
DECLARE
    lname VARCHAR2(15);
BEGIN
    SELECT last_name INTO lname FROM employees WHERE
    first_name='John';
    DBMS_OUTPUT.PUT_LINE ('John''s last name is : '
    ||lname);
EXCEPTION
    WHEN TOO_MANY_ROWS THEN
    DBMS_OUTPUT.PUT_LINE (' Команда select вернула
    несколько строк. Рассмотрите возможность
    использования курсора. ');
END;
/
```

Обработка исключений в PL/SQL

- Исключение – это ошибка PL/SQL, возникающая во время выполнения программы.
- Как возникает исключение?
 - Возбуждается неявно сервером Oracle.
 - Возбуждается явно программистом.
- Как обрабатывать исключение?
 - Перехватить с помощью обработчика исключений.
 - Распространить в вызывающую среду.

Обработка исключений



Типы исключений

- **Предопределенные, возбуждаемые сервером Oracle**
 - **Не предопределенные, возбуждаемые сервером Oracle**
 - **Пользовательские**
- Возбуждаются неявно**
- Возбуждаются явно**

Перехват исключений

Синтаксис:

```
EXCEPTION
  WHEN исключение1 [OR исключение2 . . .] THEN
    команда1;
    команда2;
    . . .
  [WHEN исключение3 [OR исключение4 . . .] THEN
    команда1;
    команда2;
    . . .]
  [WHEN OTHERS THEN
    команда1;
    команда2;
    . . .]
```


Перехват исключений: указания

- Секция обработки исключений начинается с ключевого слова `EXCEPTION`
- Допускается несколько обработчиков исключений
- Перед выходом из блока выполняется только один обработчик исключений
- Предложение `WHEN OTHERS` является последним

Перехват predefined ошибок сервера Oracle

- В программе обработки исключений ссылайтесь на стандартное имя исключения.
- Существует 20 predefined ошибок сервера БД
- Примеры predefined исключений:
 - NO_DATA_FOUND
 - TOO_MANY_ROWS
 - INVALID_CURSOR
 - ZERO_DIVIDE
 - DUP_VAL_ON_INDEX


Предопределенные ошибки сервера Oracle

Имя исключения	Код ошибки сервера Oracle	Описание
ACCESS_INTO_NULL	ORA-06530	Попытка присвоить значения атрибутам неинициализированного объекта
CASE_NOT_FOUND	ORA-06592	В выражении CASE не выбрано ни одно предложение WHEN и отсутствует предложения ELSE.
COLLECTION_IS_NULL	ORA-06531	Попытка применить методы коллекции (кроме EXISTS) , к неинициализированной вложенной таблице или элементу типа <i>varray</i> .
CURSOR_ALREADY_OPEN	ORA-06511	Попытка открыть уже открытый курсор
DUP_VAL_ON_INDEX	ORA-00001	Попытка вставки значения-дубликата
INVALID_CURSOR	ORA-01001	Недопустимая операция с курсором
INVALID_NUMBER	ORA-01722	Неудачная попытка преобразования строки символов в число
LOGIN_DENIED	ORA-01017	Попытка входа в Oracle с недействительным именем пользователя или паролем
NO_DATA_FOUND	ORA-01403	Однотрочная команда SELECT не возвращает ни одной строки
NOT_LOGGED_ON	ORA-01012	Программа PL/SQL обращается к БД, не подключившись к Oracle.
PROGRAM_ERROR	ORA-06501	Внутренняя проблема PL/SQL
ROWTYPE_MISMATCH	ORA-06504	Несовместимые типы возвращаемых данных для хост-переменной курсора и курсорной переменной PL/SQL в операции присваивания

Предопределенные ошибки сервера Oracle

Имя исключения	Код ошибки сервера	Описание
STORAGE_ERROR	ORA-06500	PL/SQL не хватает памяти или память разрушена
SUBSCRIPT_BEYOND_COUNT	ORA-06533	Номер индекса в ссылке на вложенную таблицу или элемент типа <i>varray</i> превышает количество элементов в коллекции
SUBSCRIPT_OUTSIDE_LIMIT	ORA-06532	Номер индекса в ссылке на вложенную таблицу или элемент типа <i>varray</i> выходит за пределы допустимого диапазона (например, -1).
SYS_INVALID_ROWID	ORA-01410	Неудачное преобразование строки символов в универсальный ROWID, поскольку строка символов не имеет вид правильного ROWID.
TIMEOUT_ON_RESOURCE	ORA-00051	Таймаут во время ожидания ресурса сервером Oracle
TOO_MANY_ROWS	ORA-01422	Однострочная команда SELECT возвращает более одной строки
VALUE_ERROR	ORA-06502	Ошибка в арифметической операции, операции преобразования типа данных или усечения. Возможно также нарушение ограничения размера.
ZERO_DIVIDE	ORA-01476	Попытка деления на ноль

Перехват непредопределенных исключений сервера Oracle



Непредопределенное исключение

Перехват ошибки сервера Oracle с номером -01400, невозможно вставить NULL.

```
SET SERVEROUTPUT ON
DECLARE
insert_excep EXCEPTION;
PRAGMA EXCEPTION_INIT
(insert_excep, -01400);
BEGIN
INSERT INTO departments
(department_id, department_name) VALUES (280, NULL);
EXCEPTION
WHEN insert_excep THEN
DBMS_OUTPUT.PUT_LINE('INSERT OPERATION FAILED');
DBMS_OUTPUT.PUT_LINE(SQLERRM);
END;
/
```

1

2

3

Функции для перехвата исключений


- `SQLCODE`: возвращает числовой код ошибки
- `SQLERRM`: возвращает сообщение, связанное с кодом ошибки

Функции для перехвата исключений

Пример:

```
DECLARE
    error_code      NUMBER;
    error_message   VARCHAR2 (255) ;
BEGIN
    ...
EXCEPTION
    ...
    WHEN OTHERS THEN
        ROLLBACK;
        error_code := SQLCODE ;
        error_message := SQLERRM ;
        INSERT INTO errors (e_user, e_date, error_code,
            error_message) VALUES (USER,SYSDATE,error_code,
            error_message) ;
END;
/
```


Перехват пользовательских исключений



Перехват пользовательских ИСКЛЮЧЕНИЙ

```
...
ACCEPT deptno PROMPT 'Please enter the department number: '
ACCEPT name    PROMPT 'Please enter the department name: '
DECLARE
  invalid_department EXCEPTION; ← ①
  name VARCHAR2(20) := '&name';
  deptno NUMBER := &deptno;
BEGIN
  UPDATE departments
  SET    department_name = name
  WHERE  department_id = deptno;
  IF SQL%NOTFOUND THEN
    RAISE invalid_department; ← ②
  END IF;
  COMMIT;
EXCEPTION
  WHEN invalid_department THEN ← ③
    DBMS_OUTPUT.PUT_LINE('No such department id. ');
END;
/
```

Распространение исключений, возникших в подблоке

Подблоки могут
обрабатывать
исключение или
распространять его
во внешний блок.

```
DECLARE
    . . .
    no_rows      exception;
    integrity    exception;
    PRAGMA EXCEPTION_INIT (integrity, -2292);
BEGIN
    FOR c_record IN emp_cursor LOOP
        BEGIN
            SELECT ...
            UPDATE ...
            IF SQL%NOTFOUND THEN
                RAISE no_rows;
            END IF;
        END;
    END LOOP;
EXCEPTION
    WHEN integrity THEN ...
    WHEN no_rows THEN ...
END;
/
```

Процедура RAISE_APPLICATION_ERROR

Синтаксис:

```
raise_application_error (номер_исключения,  
                         сообщение[, {TRUE | FALSE}]);
```

- Процедура, которая позволяет посылать пользовательские сообщения об исключениях из хранимых подпрограмм.
- Можно сообщать приложению об исключениях и избежать возврата необработанных исключений.

Процедура RAISE_APPLICATION_ERROR

- Может использоваться в двух местах:
 - Исполняемая секция
 - Секция исключений
- Возвращает пользователю информацию об исключении тем же способом, которым пользуется сервер Oracle при возврате необработанного исключения

RAISE_APPLICATION_ERROR

Исполняемая секция:

```
BEGIN
...
DELETE FROM employees
  WHERE manager_id = v_mgr;
IF SQL%NOTFOUND THEN
  RAISE_APPLICATION_ERROR(-20202,
    'This is not a valid manager');
END IF;
...
```

Секция обработки исключений:

```
...
EXCEPTION
  WHEN NO_DATA_FOUND THEN
    RAISE_APPLICATION_ERROR (-20201,
      'Manager is not a valid employee.');
```

END;

Итоги

- Определение исключений PL/SQL
- Добавление в блок PL/SQL секции `EXCEPTION`, предназначенной для обработки исключений во время выполнения кода
- Обработка различных типов исключений:
 - predefined, raised by the Oracle server
 - undeclared, raised by the Oracle server
 - user-defined
- Effect of exception propagation on the calling environment in nested blocks

Обзор практического занятия 8

- **Обработка именованных исключений**
- **Создание и возбуждение пользовательских исключений**